
Language Description

A language is usually described by its

• syntax (sentences of the language)

• static semantics (which sentences are wellformed, e.g.

wrt. scope rules)

• (dynamic) semantics (meaning of sentences)

• pragmatics (what is good use of the language)

Formal specification of syntax and semantics of

programming/specification languages:

• programs/specifications are modelled as mathematical

objects

• used in the design & development of

– programming languages like Chill, Ada, ...

– specification languages like VDM, RSL, Casl, Z, ...

In this course we will use RSL to specify language-based

systems.

Concrete versus Abstract Syntax

concrete syntax

Concrete syntax ∼ strings of symbols.

We use grammars (e.g. BNF) to describe concrete syntax.

Bit ::= 0 | 1

abstract syntax

By an abstract syntax we understand rules for mathematically

characterising the structure of sentences in terms of its

decomposition. Hence, an abstract syntax abstracts from the

symbols in the concrete syntax.

Abstract syntax ∼ parse trees.

We use RSL type definitions to define abstract syntax.

Bit == zero | one

Kinds of Semantics

• √
Denotational

• Mechanical/Computational/Operational

• &c.

Denotational Semantics — Principle

Every construct of the language must have its own denotation

(meaning).

The semantics is described by functions M associating a

denotation to each construct of the language. The denotation

of a construct is defined in terms of the denotations of its

subconstructs c1, c2, . . . , cn:

M(S(c1, c2, . . . , cn)) = F(M(c1),M(c2), . . . ,M(cn))

denotes

syntactic object ——–> semantic object

M

1 + 2 3

value expr value

+ addition

operator function

02263: Formal Aspects of Software Engineering

Language-based Systems

Foil 16.6

Anne Haxthausen, IMM/DTU Spring 2009

Technical University of Denmark

Informatics and Mathematical Modelling

Computer Science and Technology

IMM/DTU, Building 322, DK–2800 Lyngby, Denmark; Phone: +45-45-257510, Fax: +45-45-930007, E-mail: ah@imm.dtu.dk March 18, 2009

RSL Specification of Language-based System

specification pattern

Abstract syntax:

type /∗ type of syntactic objects ∗/
Syn = ...

Static Semantics:

value /∗ is wff(s) gives true when s is static correct ∗/
is wff : Syn → Bool

is wff(s) ≡ ...

Dynamic Semantics:

type /∗ type of semantic objects ∗/
Sem = ...

value /∗ m(s) gives the denotation of s ∗/
m : Syn ∼→ Sem

m(s) ≡ ...

pre is wff(s)

SAL1: A Simple Applicative Language

Concrete syntax examples:

5

x

e1 + e2, e1 − e2

− e1

Abstract Syntax

type

Expr = Const | InfixExpr | PrefixExpr | Id,

Const :: const : Int,

InfixExpr :: l : Expr infixop : InfixOp r : Expr,

PrefixExpr :: prefixop : PrefixOp expr : Expr,

InfixOp == plus | minus,

PrefixOp == minus,

Id :: ide : Identifier,

Identifier

Static semantics of SAL1

type /∗ dictionary of allowed identifiers ∗/
Dict = Identifier-set

value

is wff : Expr → Dict → Bool

is wff(e)(d) ≡
case e of

mk Const(i) → true,

mk Id(id) → id ∈ d,

mk InfixExpr(e1, , e2) →
is wff(e1)(d) ∧ is wff(e2)(d),

mk PrefixExpr( , e1) → is wff(e1)(d)

end

Dynamic Semantics of SAL1

type /∗ Environment ∗/ Env = Identifier →m Int

value

m : Expr → Env ∼→ Int

m(e)(ρ) ≡
case e of

mk Const(i) → i,

mk Id(id) → ρ(id),

mk InfixExpr(e1, op, e2) →
m(op)(m(e1)(ρ), m(e2)(ρ)),

mk PrefixExpr(op, e1) → m(op)(m(e1)(ρ))

end

pre is wff(e)(dom ρ),

m : InfixOp → (Int × Int → Int)

m(op) ≡
case op of

plus → λ (x, y) : Int × Int • x + y,

minus → λ (x, y) : Int × Int • x − y

end,

m : PrefixOp → (Int → Int)

m(op) ≡ (λ x : Int • 0 − x)

Alternative formulation:

value

m : Expr → Env ∼→ Int

m(e)(ρ) ≡
case e of

mk Const(i) → i,

mk Id(id) → ρ(id),

mk InfixExpr(e1, op, e2) →
case op of

plus → m(e1)(ρ) + m(e2)(ρ),

minus → m(e1)(ρ) − m(e2)(ρ)

end,

mk PrefixExpr(= minus, e1) → 0 − m(e1)(ρ)

end

pre is wff(e)(dom ρ)

Example:

m(mk InfixExpr(mk Const(1), plus, mk Const(2)))(ρ) ≡
m(plus)(m(mk Const(1))(ρ),m(mk Const(2))(ρ)) ≡
(λ (x,y) : Int × Int • x + y)(1,2) ≡
1 + 2 ≡
3

SAL2: A Simple Typed Applicative Language

Abstract Syntax (using unions and records)

type

Expr = Const | InfixExpr | PrefixExpr,

Const = IntConst | BoolConst,

IntConst :: int : Int,

BoolConst :: bool : Bool,

InfixExpr :: l : Expr infixop : InfixOp r : Expr,

PrefixExpr :: prefixop : PrefixOp ex : Expr,

InfixOp == plus | minus | and | or,

PrefixOp == minus | not

Alternative Abstract Syntax (using variants)

type

Expr ==

mk IntConst(i : Int) |
mk BoolConst(b : Bool) |
mk InfixExpr(l : Expr, infixop : InfixOp, r : Expr) |
mk PrefixExpr(prefixop : PrefixOp, ex : Expr) ,

InfixOp == plus | minus | and | or,

PrefixOp == minus | not

Static Semantics of SAL2

type Tp == integer | boolean

value

type of : Expr → Tp

type of(e) ≡
case e of

mk IntConst( ) → integer,

mk BoolConst( ) → boolean,

mk InfixExpr( , op, ) →
case op of

plus → integer,

minus → integer,

and → boolean,

or → boolean

end,

mk PrefixExpr(op, ) →
case op of

minus → integer,

not → boolean

end

end,

is wff : Expr → Bool

is wff(e) ≡
case e of

mk IntConst( ) → true,

mk BoolConst( ) → true,

mk InfixExpr(e1, op, e2) →
is wff(e1) ∧
is wff(e2) ∧
let t1 = type of(e1), t2 = type of(e2) in

case op of

plus → (t1, t2) = (integer, integer),

minus → (t1, t2) = (integer, integer),

and → (t1, t2) = (boolean, boolean),

or → (t1, t2) = (boolean, boolean)

end

end,

mk PrefixExpr(op, e1) →
is wff(e1) ∧
let t = type of(e1) in

case op of

minus → t = integer,

not → t = boolean

end

end

end

Dynamic Semantics of SAL2

Semantic domain and functions

type Val == mk Int(intg : Int) | mk Bool(bool : Bool)

value

add : Val × Val ∼→ Val

add(v1, v2) ≡ mk Int(intg(v1) + intg(v2)),

sub : Val × Val ∼→ Val

sub(v1, v2) ≡ mk Int(intg(v1) − intg(v2)),

and : Val × Val ∼→ Val

and(v1, v2) ≡ mk Bool(bool(v1) ∧ bool(v2)),

or : Val × Val ∼→ Val

or(v1, v2) ≡ mk Bool(bool(v1) ∨ bool(v2)),

minus : Val ∼→ Val

minus(v) ≡ mk Int(0 − intg(v)),

not : Val ∼→ Val

not(v) ≡ mk Bool(∼ bool(v))

Semantic meaning function

value

m : Expr ∼→ Val

m(e) ≡
case e of

mk IntConst(i) → mk Int(i),

mk BoolConst(b) → mk Bool(b),

mk InfixExpr(e1, op, e2) →
case op of

plus → add(m(e1), m(e2)),

minus → sub(m(e1), m(e2)),

and → and(m(e1), m(e2)),

or → or(m(e1), m(e2))

end,

mk PrefixExpr(op, e1) →
case op of

minus → minus(m(e1)),

not → not(m(e1))

end

end

pre is wff(e)

SIL1: A Simple Imperative Language

Concrete syntax examples

y := x + 1 ;

if b then y := 1 else y := 2 end;

x := 1 ; y := x

Abstract Syntax

type

Prog == mk Prog(stmt : Stmt),

Stmt ==

mk Asgn(ide : Identifier, expr : Expr) |
mk If(cond : Expr, s1 : Stmt, s2 : Stmt) |
mk Seq(head : Stmt, last : Stmt),

Expr ==

mk Const(const : Int) |
mk Plus(fst : Expr, snd : Expr) |
mk Id(ide : Identifier),

Identifier

Dynamic Semantics of SIL1

Applicative Model

type /∗ storage for program variables ∗/
Σ = Identifier →m Int

value

m : Prog → Σ → Σ

m(p)(σ) ≡ m(stmt(p))(σ),

m : Stmt → Σ → Σ

m(s)(σ) ≡
case s of

mk Asgn(i, e) → σ † [ i 7→ m(e)(σ) ],

mk Seq(s1, s2) → m(s2)(m(s1)(σ)),

mk If(c, s1, s2) →
if m(c)(σ) 6= 0 then m(s1)(σ) else m(s2)(σ) end

end,

m : Expr → Σ → Int

m(e)(σ) ≡
case e of

mk Const(n) → n,

mk Plus(e1, e2) → m(e1)(σ) + m(e2)(σ),

mk Id(id) → if id ∈ dom σ then σ(id) else 0 end

end

Dynamic Semantics of SIL1

Imperative Model

type Σ = Identifier →m Int

variable σ : Σ

value

m : Prog → write σ Unit

m(p) ≡ m(stmt(p)),

m : Expr → read σ Int

m(e) ≡
case e of

mk Const(n) → n,

mk Plus(e1, e2) → m(e1) + m(e2),

mk Id(id) → if id ∈ dom σ then σ(id) else 0 end

end,

m : Stmt → write σ Unit

m(s) ≡
case s of

mk Asgn(i, e) → σ := σ † [ i 7→ m(e) ],

mk Seq(s1, s2) → m(s1) ; m(s2),

mk If(c, s1, s2) → if m(c) 6= 0 then m(s1) else m(s2) end

end,

SMALL: A Small Machine Language

SMALL programs specify computations on a simple machine

consisting of a stack and a storage that stores the values of

program variables.

Abstract syntax of SMALL

type

MProg = Inst∗,
Inst ==

mk Push(ide1 : Identifier) |
mk Pop(Unit) |
mk Add(Unit) |
mk Cnst(val : Int) |
mk Store(ide2 : Identifier) |
mk Jumpfalse(off1 : Int) |
mk Jump(off2 : Int)

An interpreter for SMALL instructions

type Σ = Identifier →m Int, Stack = Int∗

value

I : MProg × Int × Stack → (Σ →Σ)

I(mp, pc, s)(σ) ≡
if pc ≤ 0 ∨ pc > len mp then σ else

case mp(pc) of

mk Push(x) → if x ∈ dom σ

then I(mp, pc + 1, 〈σ(x)〉 ̂ s)(σ)

else I(mp, pc + 1, 〈0〉 ̂ s)(σ) end,

mk Pop(()) → if len s = 0 then σ

else I(mp, pc + 1, tl s)(σ) end,

mk Cnst(n) → I(mp, pc + 1, 〈n〉 ̂ s)(σ),

mk Add(()) → if len s < 2 then σ

else I(mp, pc + 1,〈s(1) + s(2)〉 ̂ tl tl s)(σ) end,

mk Store(x) → if len s = 0 then σ

else I(mp, pc + 1, s)(σ † [ x 7→ s(1) ]) end,

mk Jumpfalse(n) → if len s = 0 then σ

elsif hd s 6= 0 then I(mp, pc + 1, s)(σ)

else I(mp, pc + n, s)(σ) end,

mk Jump(n) → I(mp, pc + n, s)(σ)

end

end

The interpretation function I operates on a program, mp, and carries with it an

instruction counter pc, the stack s and storage σ.

A Compiler for SIL1

value

comp Prog : Prog → MProg

comp Prog(p) ≡ comp Stmt(stmt(p)),

comp Stmt : Stmt → MProg

comp Stmt(s) ≡
case s of

mk Asgn(id, e) → comp Expr(e) ̂ 〈 mk Store(id), mk Pop() 〉,
mk Seq(s1, s2) → comp Stmt(s1) ̂ comp Stmt(s2),

mk If(e, s1, s2) →
let

ce = comp Expr(e),

cs1 = comp Stmt(s1), cs2 = comp Stmt(s2)

in

ce ̂

〈 mk Jumpfalse(len cs1 + 3) 〉 ̂

〈 mk Pop() 〉 ̂

cs1 ̂

〈 mk Jump(len cs2 + 2) 〉 ̂

〈 mk Pop() 〉 ̂

cs2

end

end,

comp Expr : Expr → MProg

comp Expr(e) ≡
case e of

mk Const(n) → 〈 mk Cnst(n) 〉,
mk Plus(e1, e2) →

comp Expr(e1) ̂ comp Expr(e2) ̂ 〈 mk Add() 〉,
mk Id(id) → 〈 mk Push(id) 〉

end

Compiler correctness

∀ p : Prog, σ : Σ •

I(comp Prog(p), 1, 〈〉)(σ) = m(p)(σ)

Exercise

Consider the following SIL1 program:

x := 2; if x then y := x + 3 else y := y+1 end

1. What is the corresponding abstract syntax value p?

2. What is the result (comp Prog(p)) of compiling the SIL1

program p?

3. Calculate I(comp Prog(p), 1, <>)([ x 7→ 2, y 7→ 1 ]).

4. Calculate m(p)([ x 7→ 2, y 7→ 1 ]).


